// File: contracts/Bitx.sol



// SPDX-License-Identifier: MIT
// Developed by @Rotwang9000 for Bitx.cx
// https://t.me/BitXcx
// Register for Airdrop: https://t.me/BitxLiveBot
// Buy & Stake Tokens https://token.bitx.cx

pragma solidity ^0.8.9;

import "@openzeppelin/contracts/token/ERC20/ERC20.sol";
import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/security/ReentrancyGuard.sol"; 

contract Bitx is ERC20, Ownable, ReentrancyGuard {
    

    uint256 public constant INITIAL_SUPPLY = 1000000 * (10 ** 18);  // assuming 18 decimals
    uint256 public tokensToBeSold = INITIAL_SUPPLY / 2;

    constructor() ERC20("Bitx.cx", "BITX") {
        _mint(address(this), INITIAL_SUPPLY);
    }

    // Airdrop
    mapping(address => uint16) public airdropAmounts; // This now uses uint16 to represent whole tokens up to 65,535
    mapping(address => bool) public hasClaimedAirdrop;
    uint256 constant MAX_AIRDROP_AMOUNT = 5000 * 10**18;  // example value for 18 decimals
    uint256 constant MAX_RECIPIENTS = 100; // prevent adding too many recipients at once


    // Staking
    uint256 public rewardPot = 0;
    uint256 public onDemandUnstakeFeePot;
    uint256 public totalStaked;
    uint256 public constant WAIT_PERIOD = 15 days; // Example: 30 days unlock period
    mapping(address => uint256) private accountedLastRewardRate;
    uint8 public constant NOWAIT_UNSTAKE_FEE = 8;  // 8% fee
    uint256 public gatheredRewardRate = 0; // This will represent rewards per staked token
    uint256 public constant SCALE = 9000 * 111111111111111111;


    struct StakeInfo {
        uint256 amount;
        uint256 unlockTimestamp;
    }

    mapping(address => StakeInfo) public stakers;


    // Direct Purchase
    uint256 public pricePerToken = 0.001 ether; // Example price: 1 ETH = 1000 BITX
    uint256 public bonusRate = 10; // 10% bonus per additional ETH spent
    uint256 public constant MAX_BONUS_PERCENT = 100; // e.g., a 100% max bonus

    event Purchased(address indexed user, uint256 amount);
    event Staked(address indexed user, uint256 amount);
    event Unstaked(address indexed user, uint256 amount);
    event RewardsAdded(uint256 amount);



    function addToAirdropList(address[] memory recipients, uint16[] memory amounts) external onlyOwner {
        require(recipients.length == amounts.length, "Arrays must be of equal length");
        require(recipients.length <= MAX_RECIPIENTS, "Too many recipients added at once");
        
        uint256 totalAmount = 0;
        for (uint256 i = 0; i < recipients.length; i++) {
            require(amounts[i] <= 65535, "Airdrop amount too large for a recipient");
            totalAmount = totalAmount + amounts[i];
            
            airdropAmounts[recipients[i]] = amounts[i]; // Storing as uint16 representing whole tokens
        }
        
        require(totalAmount * (10 ** decimals()) <= balanceOf(address(this)), "Not enough tokens in contract to airdrop");
    }



    function removeFromAirdropList(address[] memory recipients) external onlyOwner {
        for (uint256 i = 0; i < recipients.length; i++) {
            delete airdropAmounts[recipients[i]];
        }
    }

    function claimAirdrop() external {
        require(!hasClaimedAirdrop[msg.sender], "Airdrop already claimed");
        uint16 wholeAmount = airdropAmounts[msg.sender];
        require(wholeAmount > 0, "No airdrop amount set for caller");
        
        uint256 claimableAmount = uint256(wholeAmount) * (10 ** decimals());
        
        hasClaimedAirdrop[msg.sender] = true;
        airdropAmounts[msg.sender] = 0;
        _transfer(address(this), msg.sender, claimableAmount); // Transferring the claimable amount (with decimals)

    }


    function purchase() external payable {
        require(pricePerToken > 0, "Price per token should not be zero");

        uint256 amountToBuy = (msg.value * 10 ** decimals()) / pricePerToken;

        uint256 additionalBonusRate = (msg.value * bonusRate) / 1 ether; // Calculate the bonus rate
        if (additionalBonusRate > MAX_BONUS_PERCENT) {
            additionalBonusRate = MAX_BONUS_PERCENT; // Cap the bonus rate to MAX_BONUS_PERCENT
        }

        uint256 bonusAmount = (amountToBuy * additionalBonusRate) / 100; // Calculate the bonus amount

        amountToBuy = amountToBuy + bonusAmount;

        // Ensure that no more than 50% of the total supply can be purchased
        require(amountToBuy <= tokensToBeSold, "Not enough tokens left for sale");
        
        tokensToBeSold -= amountToBuy;

        // Transfer the tokens from the contract to the buyer
        _transfer(address(this), msg.sender, amountToBuy);
        emit Purchased(msg.sender, amountToBuy);
    }


    function stake(uint256 amount) external {
        require(amount > 0, "Amount must be greater than 0");
        require(balanceOf(msg.sender) >= amount, "Insufficient balance");

        // Automatically claim rewards for the user if any
        uint256 owedRewardRate = gatheredRewardRate - accountedLastRewardRate[msg.sender];
        uint256 reward = stakers[msg.sender].amount * owedRewardRate / SCALE;
        if (reward > 0) {
            require(address(this).balance - rewardPot >= reward, "Contract doesn't have enough Ether to reward");
            payable(msg.sender).transfer(reward);
            rewardPot -= reward;
        }


        // Update accountedLastRewardRate when user stakes
        accountedLastRewardRate[msg.sender] = gatheredRewardRate;

        // If the user increases their stake before the unlock period is over, reset the unlock timestamp.
        if (block.timestamp < stakers[msg.sender].unlockTimestamp) {
            stakers[msg.sender].unlockTimestamp = 0;
        }

        stakers[msg.sender].amount = stakers[msg.sender].amount + amount;
        totalStaked = totalStaked + amount;  // Update totalStaked when user stakes

        transferFrom(msg.sender, address(this), amount); // Transfer the staked amount to the contract
        emit Staked(msg.sender, amount);
    }

    // View function to see the staked amount for a user
    function viewStakedAmount(address _user) external view returns (uint256, uint256) {
        return (stakers[_user].amount, balanceOf(_user));
    }

    modifier hasStaked() {
        require(stakers[msg.sender].amount > 0, "No staked amount found");
        _;
    }

    function claimUnstaked() external nonReentrant hasStaked{
        require(block.timestamp >= stakers[msg.sender].unlockTimestamp, "Tokens are still locked");

        // Automatically claim rewards for the user if any
        uint256 owedRewardRate = gatheredRewardRate - accountedLastRewardRate[msg.sender];
        uint256 reward = stakers[msg.sender].amount * owedRewardRate / SCALE;
        if (reward > 0) {
            require(address(this).balance >= reward, "Contract doesn't have enough Ether to reward");
            payable(msg.sender).transfer(reward);
            rewardPot -= reward;  // Update the rewardPot after the check
            accountedLastRewardRate[msg.sender] = gatheredRewardRate;
        }

        uint256 amount = stakers[msg.sender].amount;
        stakers[msg.sender].amount = 0;
        stakers[msg.sender].unlockTimestamp = 0;

        totalStaked = totalStaked - amount;

        _transfer(address(this), msg.sender, amount); 
        emit Unstaked(msg.sender, amount);
    }



    function startUnstaking() external hasStaked {
        stakers[msg.sender].unlockTimestamp = block.timestamp + WAIT_PERIOD;
    }


    function addToRewardPot() external payable {
        require(totalStaked > 0, "Cannot distribute rewards when there are no stakers");
        uint256 rewardPerTokenScaled = (msg.value * SCALE) / totalStaked;
        gatheredRewardRate = gatheredRewardRate + rewardPerTokenScaled;
        rewardPot += msg.value; // Increase the rewardPot
        emit RewardsAdded(msg.value);
    }


    function onDemandUnstake() external hasStaked {
        StakeInfo storage stakeInfo = stakers[msg.sender];
        
        // Calculate the unclaimed rewards for this unstaker
        uint256 owedRewardRate = gatheredRewardRate - accountedLastRewardRate[msg.sender];
        uint256 rewardToRedistribute = (stakeInfo.amount * owedRewardRate) / SCALE;
        
        // Calculate the fee for instant unstaking
        uint256 fee = (stakeInfo.amount * NOWAIT_UNSTAKE_FEE) / 100;
        uint256 refundAmount = stakeInfo.amount - fee;
        
        // Update totalStaked and other state variables
        totalStaked = totalStaked - stakeInfo.amount;
        stakeInfo.amount = 0;
        stakeInfo.unlockTimestamp = 0;
        
        // Redistribute the unclaimed rewards to the remaining stakers
        if (totalStaked > 0) {
            uint256 rewardPerTokenScaled = (rewardToRedistribute * SCALE) / totalStaked;
            gatheredRewardRate += rewardPerTokenScaled;
        }
        
        onDemandUnstakeFeePot = onDemandUnstakeFeePot + fee;

        _transfer(address(this), msg.sender, refundAmount);
    }

    // Function to withdraw and reset onDemandUnstakeFeePot
    function withdrawOnDemandUnstakeFeePot() external onlyOwner {
        uint256 amount = onDemandUnstakeFeePot;
        require(amount > 0, "No funds to withdraw");
        
        onDemandUnstakeFeePot = 0; // Resetting the pot
        
        _transfer(address(this), msg.sender, amount); // Transfer to owner
    }


    function claimRewards() external nonReentrant hasStaked{  
        uint256 owedRewardRate = gatheredRewardRate - accountedLastRewardRate[msg.sender];
        uint256 reward = stakers[msg.sender].amount * owedRewardRate / SCALE;

        require(reward > 0, "No rewards available");
        require(address(this).balance >= reward, "Contract doesn't have enough Ether to reward");

        payable(msg.sender).transfer(reward); 
        rewardPot -= reward;

        // Ensure the rewardPot has not underflowed
        assert(rewardPot <= address(this).balance); 

        accountedLastRewardRate[msg.sender] = gatheredRewardRate;
    }

    // View function to see the reward
    function viewRewards(address _user) external view returns (uint256) {
        uint256 owedRewardRate = gatheredRewardRate - accountedLastRewardRate[_user];
        return stakers[_user].amount * owedRewardRate/SCALE;
    }

    // Modify withdrawETH to only allow withdrawal of spare Ether
    function withdrawETH() external onlyOwner {
        uint256 spareETH = address(this).balance - rewardPot;
        payable(owner()).transfer(spareETH);
    }

    function setPricePerToken(uint256 newPrice) external onlyOwner {
        require(newPrice > 0, "Price cannot be zero");
        pricePerToken = newPrice;
    }

    function setBonusRate(uint256 newRate) external onlyOwner {
        require(newRate <= 100, "Bonus rate should not be more than 100%");
        bonusRate = newRate;
    }

    function getPriceAndBonus() external view returns (uint256, uint256) {
        return (pricePerToken, bonusRate);
    }

}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Returns true if the reentrancy guard is currently set to "entered", which indicates there is a
     * `nonReentrant` function in the call stack.
     */
    function _reentrancyGuardEntered() internal view returns (bool) {
        return _status == _ENTERED;
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC20/ERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/ERC20.sol)

pragma solidity ^0.8.0;

import "./IERC20.sol";
import "./extensions/IERC20Metadata.sol";
import "../../utils/Context.sol";

/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.openzeppelin.com/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * The default value of {decimals} is 18. To change this, you should override
 * this function so it returns a different value.
 *
 * We have followed general OpenZeppelin Contracts guidelines: functions revert
 * instead returning `false` on failure. This behavior is nonetheless
 * conventional and does not conflict with the expectations of ERC20
 * applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20 is Context, IERC20, IERC20Metadata {
    mapping(address => uint256) private _balances;

    mapping(address => mapping(address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;

    /**
     * @dev Sets the values for {name} and {symbol}.
     *
     * All two of these values are immutable: they can only be set once during
     * construction.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5.05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the default value returned by this function, unless
     * it's overridden.
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view virtual override returns (uint8) {
        return 18;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view virtual override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address to, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _transfer(owner, to, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * NOTE: If `amount` is the maximum `uint256`, the allowance is not updated on
     * `transferFrom`. This is semantically equivalent to an infinite approval.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20}.
     *
     * NOTE: Does not update the allowance if the current allowance
     * is the maximum `uint256`.
     *
     * Requirements:
     *
     * - `from` and `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     * - the caller must have allowance for ``from``'s tokens of at least
     * `amount`.
     */
    function transferFrom(address from, address to, uint256 amount) public virtual override returns (bool) {
        address spender = _msgSender();
        _spendAllowance(from, spender, amount);
        _transfer(from, to, amount);
        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, allowance(owner, spender) + addedValue);
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        address owner = _msgSender();
        uint256 currentAllowance = allowance(owner, spender);
        require(currentAllowance >= subtractedValue, "ERC20: decreased allowance below zero");
        unchecked {
            _approve(owner, spender, currentAllowance - subtractedValue);
        }

        return true;
    }

    /**
     * @dev Moves `amount` of tokens from `from` to `to`.
     *
     * This internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     */
    function _transfer(address from, address to, uint256 amount) internal virtual {
        require(from != address(0), "ERC20: transfer from the zero address");
        require(to != address(0), "ERC20: transfer to the zero address");

        _beforeTokenTransfer(from, to, amount);

        uint256 fromBalance = _balances[from];
        require(fromBalance >= amount, "ERC20: transfer amount exceeds balance");
        unchecked {
            _balances[from] = fromBalance - amount;
            // Overflow not possible: the sum of all balances is capped by totalSupply, and the sum is preserved by
            // decrementing then incrementing.
            _balances[to] += amount;
        }

        emit Transfer(from, to, amount);

        _afterTokenTransfer(from, to, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _beforeTokenTransfer(address(0), account, amount);

        _totalSupply += amount;
        unchecked {
            // Overflow not possible: balance + amount is at most totalSupply + amount, which is checked above.
            _balances[account] += amount;
        }
        emit Transfer(address(0), account, amount);

        _afterTokenTransfer(address(0), account, amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, reducing the
     * total supply.
     *
     * Emits a {Transfer} event with `to` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     * - `account` must have at least `amount` tokens.
     */
    function _burn(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: burn from the zero address");

        _beforeTokenTransfer(account, address(0), amount);

        uint256 accountBalance = _balances[account];
        require(accountBalance >= amount, "ERC20: burn amount exceeds balance");
        unchecked {
            _balances[account] = accountBalance - amount;
            // Overflow not possible: amount <= accountBalance <= totalSupply.
            _totalSupply -= amount;
        }

        emit Transfer(account, address(0), amount);

        _afterTokenTransfer(account, address(0), amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(address owner, address spender, uint256 amount) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Updates `owner` s allowance for `spender` based on spent `amount`.
     *
     * Does not update the allowance amount in case of infinite allowance.
     * Revert if not enough allowance is available.
     *
     * Might emit an {Approval} event.
     */
    function _spendAllowance(address owner, address spender, uint256 amount) internal virtual {
        uint256 currentAllowance = allowance(owner, spender);
        if (currentAllowance != type(uint256).max) {
            require(currentAllowance >= amount, "ERC20: insufficient allowance");
            unchecked {
                _approve(owner, spender, currentAllowance - amount);
            }
        }
    }

    /**
     * @dev Hook that is called before any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * will be transferred to `to`.
     * - when `from` is zero, `amount` tokens will be minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(address from, address to, uint256 amount) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * has been transferred to `to`.
     * - when `from` is zero, `amount` tokens have been minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens have been burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(address from, address to, uint256 amount) internal virtual {}
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/IERC20Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";

/**
 * @dev Interface for the optional metadata functions from the ERC20 standard.
 *
 * _Available since v4.1._
 */
interface IERC20Metadata is IERC20 {
    /**
     * @dev Returns the name of the token.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the symbol of the token.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the decimals places of the token.
     */
    function decimals() external view returns (uint8);
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}